
9. PRZYKŁAD OPROGRAMOWANIA MES  

9.1.  SFORMUŁOWANIE PROBLEMU 

W ostatnim rozdziale przedstawimy przykład oprogramowania, rozwiązującego dwu-
wymiarowy problem nieustalonej wymiany ciepła. Podstawy teoretyczne tego rozwiązania 
są podane w rozdziałach 4 i 5. Układ równani MES moŜna uzyskać za pomocą formuły 
(5.8): 

gdzie składniki formuły (5.8) moŜna obliczyć za pomocą formuł (4.14), (4.15) i (5.3): 

W formule (5.8) { }1t  - wartości temperatur węzłowych po czasie τ∆ , wektor { }0t  

reprezentuje temperatury węzłowe w chwili 0=τ .  
PoniewaŜ analogiczną formę mają równania dla zadani dyfuzji, opracowany program 

po zmianie współczynników moŜna wykorzystać do modelowaniu procesów dyfuzji. Wy-
zerowanie matryca [C] spowoduje rozwiązanie ustalonego zadania cieplnego, albo innego, 
opartego o równanie Poissona (6.65).  

9.2.  DANE WEJŚCIOWE 

Dane wejściowe do opracowanego programu są zadawane w postaci pliku tekstowego. 
Przykładową zawartość takiego pliku, z opisem wczytywanych zmiennych, pokazano na 
rys. 9.1.  

 
100     mTbegin        Temperatura początkowa, C 
1000    mTime          Czas procesu, s 
1       mdTime           Krok czasowy, s 
1200    mT_otoczenia   Temperatura otoczenia C 
300     mAlfa           Współczynnik wymiany ciepła  W/Cm2 
100     mH0             Wysokość przekroju, mm 
100     mB0             Szerokość przekroju, mm 
25      mNhH           Liczba węzłów po wysokości 
25      mNhH            Liczba węzłów po szerokości 
700     mC              Ciepło właściwe, J/Ckg 
25      mK              Współczynnik przewodzenia ciepła 
7800    mR              Gęstość kg/m3 

Rys. 9.1. Przykład zawartości pliku z danymi do obliczeń.  

9.3.  KOD ŹRÓDŁOWY PROGRAMU   

Program zawiera kilku modułów, zawartość najwaŜniejszych z nich jest opisana poni-
Ŝej. Pełny tekst programu w języku FORTRAN90 moŜna znaleźć na stronie internetowej 
http://home.agh.edu.pl/~milenin/programs/.   

9.3.1. PROGRAM GŁÓWNY TEMP2D 

Listing głównego programu TEMP2D jest podany na rys. 9.2. Oznaczenia głównych 
zmiennych są następujące. 

GlobData – nazwa modułu z opisem globalnych zmiennych. 
dTauMax – maksymalnie dopuszczalny krok czasowy, obliczony z warunku nume-

rycznej stabilności rozwiązania. 
n, Ntau – numer bieŜącego kroku czasowego i ilość kroków czasowych.  
Oznaczenia głównych procedur, wyzywanych w programie głównym są następujące. 
IniEL4 – procedura inicjalizacji danych dla wybranego typu elementu skończonego 

(wybrany typ: 4-węzłowy element, opisany w rozdziale (3.9)).  
InpData – wczytywanie dane wyjściowych. 
GenGrid2d – rozbudowa siatki MES dla obszaru o przekroju prostokątnym.  
SetControlPoints – definicja punktów, z jakich będzie wyprowadzane do pliku teksto-

wego wyniki rozwiązania.  
ALLOCATE_Matrix – przydzielenie pamięci dla tablic. 
WriteControlPointsBegin – zapis do pliku z wynikami obliczeń danych wstępnych. 

WriteControlPoints – zapis do pliku wyników obliczeń. 
SOLVER – procedura rozwiązywania zadania dla bieŜącego kroku czasowego.  
 

!************************************************************************* 
!  PROGRAM: TEMP2D 
!  Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
!************************************************************************ 

program TEMP2D 
 use GlobData; 
 implicit none 
 real*8    :: Asr,dTauMax,TauP; 
 integer*4 :: n, Ntau, i, iErr; 
 
call IniEL4();      
call InpData();      
call GenGrid2d(mH0,mB0,mNhH,mNhB, mGr);  
call SetControlPoints();     
CALL ALLOCATE_Matrix();    
CALL WriteControlPointsBegin(); 
 
 Asr  = mK / (mC * mR); 
 mdTime = DBLE(mB0/(1e3*mNhB))**2/(0.5*Asr); 
CALL WriteControlPoints(); 
 Ntau = int(mTime/mdTime)+1; 
 mdTime = mTime/DBLE(Ntau); 
 mTau    = 0; 
 do n=1, Ntau 
  mTau = mTau + mdTime; 
  CALL SOLVER(  ); 
  CALL WriteControlPoints(); 
 end do; 
 CLOSE(88); CLOSE(89); 
 DEALLOCATE ( mA, STAT=iErr ); 
 DEALLOCATE ( mB, STAT=iErr ); 
 DEALLOCATE ( mX, STAT=iErr ); 
end program TEMP2D 

Rys. 9.2. Główny program TEMP2D.  

9.3.2. MODUŁ Z DEFINICJĄ STRUKTUR DANYCH  

Wykorzystanie struktur danych ułatwia proces programowania i analizy kodu. Na rys. 
9.3 pokazano listing modułu My_typ z definicją struktur danych. Wykorzystano następują-
ce struktury.  

Node – struktura danych, zawierająca informacje o węzie siatki: x, y – współrzędne 
węzła, t – temperatura w węźle, CR – prędkość zmiany temperatury.  

Element – struktura danych, zawierająca informacje o elemencie siatki MES: nop – 
mapa elementu (rozdział 7.21), Npov – liczba powierzchni elementu, kontaktujących z 
otoczeniem; aPov – lokalne numery kontaktowych powierzchni elementu.  

gr2d – struktura danych, zawierająca siatkę MES: nh – liczba węzłów; ne – liczba 
elementów; nbn – liczba węzłów w wykorzystywanym elemencie skończonym; ncn – licz-
ba zmiennych w elemencie skończonym, dla zadani cieplnych odpowiada nbn; nhPov – 
liczba węzłów siatki, znajdujących się na powierzchni rozpatrywanego obszaru; EL – tabli-
ca długością ne, zawierająca dane typu Element dla kaŜdego elementu siatki; ND – tablica 
długością nh, zawierająca dane typu Node dla kaŜdego węzła siatki. 

cor_L – struktura danych, zawierająca współrzędne N, E punktu w układzie lokalnym 
elementu. 

Gran – struktura danych, zawierająca dane o graniach powierzchniowych wykorzy-
stywanego elementu skończonego: N_p – liczba punktów całkowania po powierzchni gra-
ni; P – tablica elementów typu cor_L długością N_p zawierająca współrzędne lokalne 
punktów całkowania Gaussa; W – tablica długością N_p zawierająca wagę punktów całko-
wania odpowiednio formule (3.31); Nf – tablica o wymiarach nbn, N_p zawierająca warto-
ści funkcji kształtu elementu, obliczone w punktach całkowania na powierzchni grani; 
N1,N2 – tablicy o wymiarach nbn, N_p zawierające wartości pochodnych funkcji kształtu 
względem współrzędnych lokalnych, obliczone w punktach całkowania na powierzchni 
grani; UZEL – tablica zawierająca lokalne numery węzłów grani.  

ELEM – struktura danych, zawierająca dane o wykorzystywanym elemencie skończo-
nym: nbn – liczba węzłów w wykorzystywanym elemencie skończonym, nbnp – liczba 
węzłów w wykorzystywanym elemencie skończonym do interpolacji średniego napręŜenia 
(wykorzystywana w zadaniach teorii plastyczności), N_p – liczba punktów całkowania w 
objętości elementu; Nf – tablica o wymiarach nbn, N_p zawierająca wartości funkcji kształ-
tu elementu, obliczone w punktach całkowania w objętości elementu; N1, N2 – tablicy o 
wymiarach nbn, N_p zawierające wartości pochodnych funkcji kształtu względem współ-
rzędnych lokalnych, obliczone w punktach całkowania w objętości wykorzystywanego 
elementu sączonego; P – tablica elementów typu cor_L długością N_p zawierająca współ-
rzędne lokalne punktów całkowania Gaussa w objętości elementu; L – tablica elementów 
typu cor_L długością N_p zawierająca współrzędne lokalne węzłów elementu; W – tablica 
długością N_p zawierająca wagę punktów całkowania w objętości elementu odpowiednio 
formule (3.31); Sf – tablica długością równą maksymalnej liczbie powierzchni wykorzysta-
nego elementu skończonego (dla wykorzystywanego elementu równa 4), zawierająca ele-
menty typu Gran.  

 
 
! ****************************************************************** 
! Typy danych 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
module My_typ; 
implicit none; 
 
 type node 
  real*8 :: x,y,t,CR; 
  integer*4 :: status; 
 end type 
 
 type element 
 integer*4 :: nop(4); 

 integer*4 :: Npov;     ! Liczba kontaktowych powierzchni elementu 0-1-2 
 integer*4 :: aPov(2);  !  Lokalne numery kontaktowych powierzchni elementu 
 end type 
 
 type gr2d 
  integer*4 :: nh, ne, nbn, ncn, nhPov; 
  type(element) , dimension(:), pointer :: EL; 
  type(node) , dimension(:), pointer :: ND; 
 end type 
 
 type cor_L; 

real*8 N, E; 
 end type cor_L; 
 
 type Gran 
   integer*4 :: N_p;  ! Liczba punktów całkowania po powierzchni grani  
   type(cor_L), dimension(:), pointer :: P;  
    real*8, dimension(:), pointer      :: W;  
    real*8, dimension(:,:), pointer    :: N1,N2,Nf;  ! N1(nbn,N_p) 
    integer*1, dimension(:),pointer :: UZEL; 
  ! Lokalne numery węzłów grani 
 end type Gran; 
 

type ELEM; 
 integer*4 :: nbn, nbnp, N_p; 
 real*8, dimension(:,:), pointer :: N1,N2,Nf;  ! N1(nbn,N_p) 
 type(cor_L), dimension(:), pointer :: P;  ! P(N_p) 
 type(cor_L), dimension(:), pointer :: L; ! Współrzędne węzłów elementu 
 real*8, dimension(:), pointer :: W; ! Waga punktów całkowania W(N_p) 
 type(Gran) :: Sf(4); 

 end type ELEM; 
 
end module My_typ; 

Rys. 9.3. Typy (struktury) danych, potrzebne do rozwiązania.  

9.3.3. MODUŁ Z DEFINICJĄ ZMIENNYCH  GLOBALNYCH  

Moduł GlobData zawiera definicje zmiennych, widocznych we wszystkich procedu-
rach, gdzie zadeklarowano wykorzystanie tego modułu: use GlobData. Dokładny opis 
zmiennych podano na rys. 9.4.  

module GlobData; 
! ****************************************************************** 
! Globalne zmianne 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
use My_typ; 
implicit none; 
 real*8       ::  mTbegin;  ! Początkowa temperatura 
 real*8       ::  mTime;  ! Czas procesu 

 real*8       ::  mdTime;  ! Początkowa wartość przyrostu czasu 
 real*8       ::  mTau;  ! BieŜący czas 
 real*8       ::  mT_otoczenia; ! Temperatura otoczenia 
 real*8       ::  mAlfa;  ! Współczynnik wymiany ciepła 
 real*8       ::  mH0;  ! Wysokość przekroju 
 real*8       ::  mB0;  ! Szerokość przekroju 
 integer*4    ::  mNhH;  ! Liczba węzłów na wysokości przekroju  
 integer*4    ::  mNhB;  ! Liczba węzłów na szerokości przekroju  
 integer*4    ::  mLDA;  ! Szerokość pasma matrycy MES  
 real*8       ::  mC;   ! Pojemność cieplna 
 real*8       ::  mK;   ! Współczynnik przewodzenia ciepła 
 real*8       ::  mR;   ! Gęstość 
 type(gr2d)   :: mGr;  ! Siatka MES 
 type(ELEM)   :: mEL4;  ! Dane uszytego elementu skończonego 
 integer*4 :: mContrPoints(9); ! Kontrolne punkty 
 REAL*8    :: mcpX(9),mcpY(9);  
 real*8, dimension(4,4) :: est;   ! Matryca bieŜącego elementu 
 real*8, dimension(4)   :: r;     ! Wektor obciąŜeń bieŜącego elementu 
 real*8, dimension(:),   allocatable :: mB; ! Globalny wektor obciąŜeń 
 real*8, dimension(:,:), allocatable :: mA; ! Globalna matryca 
 real*8, dimension(:),   allocatable :: mX; ! Globalny wektor niewiadomych 
end module GlobData; 

Rys. 9.4. Globalne zmienne, wykorzystywane w programie.  

 

9.3.4. PROCEDURA INICJALIZACJI DANYCH DLA ELEMENTU   

Wypełnienie zmiennej mEL4 typu ELEM odbywa się w programie IniEL4, pokazanej 
na rys. 9.5. W pierwszej części programu zadajemy liczbę węzłów elementu nbn równą 4. 
Wybrany schemat całkowania numerycznego zawiera 4 punkty odpowiednio rys. 3.19, 
dlatego zadajemy N_p=4. Dalej przedzielamy pamięć tablicom N1, N2, Nf, P, W oraz L. 
Dalej wypełniamy współrzędne lokalne punktów całkowania odpowiednio rys. 3.19. Na-
stępnie obliczamy w kaŜdym punkcie całkowania wartości N1, N2, Nf, wykorzystując 
formuły (3.9): 

Analogiczne wypełniamy struktury dla 4 powierzchni wybranego elementu.  

Przy wykorzystaniu innego typu elementów skończonych potrzebuje zmiany jedynie 
struktura mEL4 i sama siatka elementów.  

 
! ****************************************************************** 
! Inicjalizacja danych dla elementu 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
SUBROUTINE IniEL4(); 
use GlobData; 
implicit none; 
REAL*8    :: Alfa,L1,L2,e,n,SN; 
integer*4 :: iP,i; 
 
 Alfa = 1/(sqrt(3.0)); 
 mEL4%nbn=4;  
 mEL4%N_p=4;  
 ALLOCATE(mEL4%N1( mEL4%nbn, mEL4%N_p )); 
 ALLOCATE(mEL4%N2( mEL4%nbn, mEL4%N_p )); 
 ALLOCATE(mEL4%Nf( mEL4%nbn, mEL4%N_p )); 
 ALLOCATE(mEL4%P ( mEL4%N_p )); 
 ALLOCATE(mEL4%W ( mEL4%N_p )); 
 ALLOCATE(mEL4%L ( mEL4%nbn )); 
 mEL4%W = 1.0; 
 mEL4%P(1)%E = -Alfa; 
 mEL4%P(1)%N = -Alfa; 
 mEL4%P(2)%E =  Alfa; 
 mEL4%P(2)%N = -Alfa; 
 mEL4%P(3)%E =  Alfa; 
 mEL4%P(3)%N =  Alfa; 
 mEL4%P(4)%E = -Alfa; 
 mEL4%P(4)%N =  Alfa; 
 mEL4%L(1)%E = -1; 
 mEL4%L(1)%N = -1; 
 mEL4%L(2)%E =  1; 
 mEL4%L(2)%N = -1; 
 mEL4%L(3)%E =  1; 
 mEL4%L(3)%N =  1; 
 mEL4%L(4)%E = -1; 
 mEL4%L(4)%N =  1; 
 DO iP = 1, mEL4%N_p 
  L1 = mEL4%P(iP)%E 
  L2 = mEL4%P(iP)%N 
  mEL4%Nf(1,iP) =  0.25*(1-L1)*(1-L2); 
  mEL4%N1(1,iP) = -0.25*(1-L2); 
  mEL4%N2(1,iP) = -0.25*(1-L1); 
  mEL4%Nf(2,iP) =  0.25*(1+L1)*(1-L2); 
  mEL4%N1(2,iP) =  0.25*(1-L2); 
  mEL4%N2(2,iP) = -0.25*(1+L1);  
  mEL4%Nf(3,iP) =  0.25*(1+L1)*(1+L2); 
  mEL4%N1(3,iP) =  0.25*(1+L2); 
  mEL4%N2(3,iP) =  0.25*(1+L1); 

  mEL4%Nf(4,iP) =  0.25*(1-L1)*(1+L2); 
  mEL4%N1(4,iP) = -0.25*(1+L2); 
  mEL4%N2(4,iP) =  0.25*(1-L1); 
 end do; 
 
 DO i = 1, 4 
  mEL4%Sf(i)%N_p = 2; 
  ALLOCATE( mEL4%Sf(i)%P( mEL4%Sf(i)%N_p ) ); 
  ALLOCATE( mEL4%Sf(i)%W( mEL4%Sf(i)%N_p ) ); 
  mEL4%Sf(i)%W = 1; 
  ALLOCATE( mEL4%Sf(i)%N1( mEL4%nbn, mEL4%Sf(i)%N_p)); 
  ALLOCATE( mEL4%Sf(i)%N2( mEL4%nbn, mEL4%Sf(i)%N_p)); 
  ALLOCATE( mEL4%Sf(i)%Nf( mEL4%nbn, mEL4%Sf(i)%N_p)); 
  ALLOCATE( mEL4%Sf(i)%UZEL( 2 ) ); 
 END DO; 
 
 mEL4%Sf(1)%UZEL(1) = 4; mEL4%Sf(1)%UZEL(2) = 1; 
 mEL4%Sf(2)%UZEL(1) = 1; mEL4%Sf(2)%UZEL(2) = 2; 
 mEL4%Sf(3)%UZEL(1) = 2; mEL4%Sf(3)%UZEL(2) = 3; 
 mEL4%Sf(4)%UZEL(1) = 3; mEL4%Sf(4)%UZEL(2) = 4; 
 mEL4%Sf(1)%P(1)%N=Alfa; mEL4%Sf(1)%P(1)%E=-1; 
 mEL4%Sf(1)%P(2)%N=-Alfa; mEL4%Sf(1)%P(2)%E=-1; 
 mEL4%Sf(2)%P(1)%N=-1;  mEL4%Sf(2)%P(1)%E=-Alfa; 
 mEL4%Sf(2)%P(2)%N=-1;  mEL4%Sf(2)%P(2)%E=Alfa; 
 mEL4%Sf(3)%P(1)%N=-Alfa; mEL4%Sf(3)%P(1)%E=1; 
 mEL4%Sf(3)%P(2)%N=Alfa; mEL4%Sf(3)%P(2)%E=1; 
 mEL4%Sf(4)%P(1)%N=1;  mEL4%Sf(4)%P(1)%E=Alfa; 
 mEL4%Sf(4)%P(2)%N=1;  mEL4%Sf(4)%P(2)%E=-Alfa; 
 
 do i = 1, 4 
  DO iP = 1, 2 
   e = mEL4%Sf(i)%P(ip)%E;  
   n = mEL4%Sf(i)%P(ip)%N;  
   mEL4%Sf(i)%Nf(1,ip) = 0.25*(1-e)*(1-n); 
   mEL4%Sf(i)%Nf(2,ip) = 0.25*(1+e)*(1-n); 
   mEL4%Sf(i)%Nf(3,ip) = 0.25*(1+e)*(1+n); 
   mEL4%Sf(i)%Nf(4,ip) = 0.25*(1-e)*(1+n); 
   mEL4%Sf(i)%N1(1,ip) = -0.25*(1-n); 
   mEL4%Sf(i)%N1(2,ip) = 0.25*(1-n); 
   mEL4%Sf(i)%N1(3,ip) = 0.25*(1+n); 
   mEL4%Sf(i)%N1(4,ip) = -0.25*(1+n); 
   mEL4%Sf(i)%N2(1,ip) = -0.25*(1-e); 
   mEL4%Sf(i)%N2(2,ip) = -0.25*(1+e); 
   mEL4%Sf(i)%N2(3,ip) = 0.25*(1+e); 
   mEL4%Sf(i)%N2(4,ip) = 0.25*(1-e); 
  END DO; 
 end do; 
END SUBROUTINE IniEL4; 

Rys. 9.5. Inicjalizacja danych dla czworościennego elementu.  

9.3.5. PROCEDURA WCZYTYWANIA DANYCH    

Program wczytywania danych jest pokazany na rys. 9.6. On otwiera tekstowy plik in-
data.t2d, struktura jakiego opisana w 9.2. 

 
! ****************************************************************** 
! Wczytywanie danych 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
Subroutine InpData(); 
use GlobData; 
implicit none; 
integer nFile,i,j,iost; 
 
 nFile = 89 
 OPEN (nFile, FILE='indata.t2d', action='read') 
 read(nFile, *); 
 read(nFile, *); 
 read(nFile, *); 
 read(nFile, *); 
 read(nFile, *) mTbegin; 
 read(nFile, *) mTime; 
 read(nFile, *) mdTime; 
 read(nFile, *) mT_otoczenia; 
 read(nFile, *) mAlfa; 
 read(nFile, *) mH0; 
 read(nFile, *) mB0; 
 read(nFile, *) mNhH; 
 read(nFile, *) mNhB; 
 read(nFile, *) mC; 
 read(nFile, *) mK; 
 read(nFile, *) mR; 
 close (nFile); 
end Subroutine InpData 

Rys. 9.6 Program wczytywania danych.  

9.3.6. PROCEDURA PRZYDZIELENIA PAMIĘCI DLA TABLIC  

Przydzielenie pamięci dla tablic oblicza szerokość mLDA matrycy sztywności za po-
mocą mapy elementów i przydziela pamięci zmiennym mA (globalna matryca sztywności), 
mX (wektor niewiadomych), mB (wektor obciąŜeń). W celach metodycznych pokazano, 
jak przydzielić pamięć dla pełnej matrycy o wymiarach nh*nh. Taki wariant, jednak, po-
woduje zbyt nieskuteczne wykorzystanie pamięci.    

 
SUBROUTINE ALLOCATE_Matrix; 
use GlobData; 
implicit none; 
integer*4 :: NEL, i, j, ii, jj, jB, nk(4),NeMaxB, ierr; 

 mLDA=0; 
 do NEL = 1,mGr%ne 
  do i=1,mEL4%nbn 
   nk(i) = mGr%EL(NEL)%nop(i); 
  end do; 
  do i=1,mEL4%nbn 

! * ii - wiersz * 
   ii=nk(i);  
   do j=1,mEL4%nbn  
   ! * jj - kolumna *  
    jj=nk(j); 
    jB = jj-ii+1; 
    if(jB>=mLDA) then 
    mLDA=jB; 
    NeMaxB = NEL; 
    end if; 
   end do; 
  end do; 
 end do;  
 
 ALLOCATE ( mA(mLDA, mGr%nh), STAT=iErr );  
! ALLOCATE ( mA(mGr%nh, mGr%nh), STAT=iErr );  
!    Pelna matryca MES  
 ALLOCATE ( mB(mGr%nh), STAT=iErr ); 
 ALLOCATE ( mX(mGr%nh), STAT=iErr ); 
END SUBROUTINE ALLOCATE_Matrix 

Rys. 9.7. Program do obliczenia potrzebnej pamięci dla globalnej matrycy sztywności.  

9.3.7. PROCEDURA ROZWIĄZYWANIA ZADANIA DLA 
BIEśĄCEGO KROKU CZASOWEGO  

Program do rozwiązywania zadania dla bieŜącego kroku czasowego jest pokazany na 
rys. 9.8. W tym programie oznaczenia zmiennych są następujący:  

NEL – bieŜący numer elementu; 
Nk – tablica, zawierającą mapę bieŜącego elementu; 
FeSM_heat – program wypełnienia matrycy i wektora obciąŜeń bieŜącego elementu 

NEL;  
DLSAQS – program rozwiązania układu równań z pasmową matrycą; 
DLSARG – program rozwiązania układu równań z pełną matrycą (podano jako ko-

mentarz w celach metodycznych). 
 
! ****************************************************************** 
!             Solver 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
SUBROUTINE SOLVER( ); 
use GlobData; 
use msimsl; 

implicit none; 
integer*4  :: NEL; 
integer*4, dimension(4)   :: nk; 
integer*4 :: jB,iB,NeMaxB,NCODA,i,j,ii,jj; 
integer*4 :: iErr; 
 
 mA=0;  
 mB=0;  
 mX=0; 
 do NEL=1, mGr%ne 
  do i=1, mEL4%nbn 
   nk(i) = mGr%EL(NEL)%nop(i); 
  end do; 
 
  CALL FeSM_heat( NEL ); 
 
  do i=1, mEL4%nbn 
   ii = nk(i);   ! wiersz w pełnej matryce 
   do j=1, mEL4%nbn 
    jj = nk(j);  ! Kolumna w pełnej matryce  
    iB = mLDA + ii - jj;! Wiersz w pasmowej matryce 
    if ((jj>=ii).AND.(iB<=mLDA)) then 
  mA(iB, jj) = mA(iB, jj) + est(i,j);  ! Wypełnienie pasmowej matrycy 
    end if;             
 ! mA(ii,jj) = mA(ii,jj) + est(i,j);  ! Wypełnienie pełnej matrycy  
   end do; 
   mB(ii) = mB(ii) + r(i); 
  end do; 
 end do; 
 
 NCODA = mLDA-1; 
 CALL DLSAQS (mGr%nh, mA, mLDA, NCODA, mB, mX) 
! Program rozwiązania układu równań z pasmowej matrycą  
! NCODA=1; 
! CALL DLSARG (mGr%nh, mA, mGr%nh, mB, NCODA, mX) 
! Program rozwiązania układu równań z pełnej matrycą  
 do i=1, mGr%nh 
  mGr%nd(i)%CR = ( mGr%nd(i)%t-mX(i) ) / mdTime; 
  mGr%nd(i)%t = mX(i); 
 end do; 
END SUBROUTINE SOLVER; 

Rys. 9.8. Program wypełnienia globalnej matrycy i rozwiązania układu równani MES.  

9.3.8. PROCEDURA WYPEŁNIENIA MACIERZY I WEKTORA 
OBCIĄśEŃ ELEMENTU 

Program FeSM_heat (rys. 9.9) zawiera dwie części:  
PRE_heat_mat –  wypełnia matryce i wektor obciąŜeń;  
PRE_heat_pov_mat – uwzględnia warunki brzegowe. 

 191 

 
! ****************************************************************** 
! Generacja matryc elementu NEL 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! ****************************************************************** 
SUBROUTINE FeSM_heat( NEL ); 
use GlobData; 
implicit none; 
integer*4  :: NEL; ! Numer bieŜącego Elementu  
 est = 0;   
 r = 0; 
 call PRE_heat_mat( NEL ); 
 call PRE_heat_pov_mat(NEL ); 
END SUBROUTINE FeSM_heat; 

Rys. 9.9. Program do obliczenia matrycy sztywności bieŜącego elementu NEL.  

9.3.9. PROCEDURA WYPEŁNIENIA MACIERZY I WEKTORA 
OBCIĄśEŃ 

Program jest podany na rys. 9.10. Wywołuje program Jacob_2d, opisany w rozdziale 
3.6 i podany w postaci uproszczonej na rys. 3.26. Program realizuje typowy algorytm wy-
pełnienia matrycy jednego elementu, wielokrotnie opisany w danym opracowaniu. W tek-
ście programu podane są odwołania na formuły, wykorzystywane podczas obliczeń elemen-
tów matrycy.    

 
subroutine PRE_heat_mat( NEL ); 
use GlobData; 
implicit none; 
integer*4  :: NEL;  
INTEGER*4  :: I,j,N,P,Id; 
REAL*8, DIMENSION(2,2) :: J_,J_inv; ! Matryca Jakobiego i odwrotna matryca 
REAL*8    :: DetJ,Ni,Nn,Hin,Cin; 
real*8    :: T0p; 
REAL*8, DIMENSION(4)   :: Ndx,Ndy;  ! Pochodny funkcji kształtu względem X i Y 
REAL*8, DIMENSION(4)   :: X,Y,Temp_0;   
 
 DO I=1,mEL4%NBN 
  Id = ABS( mGr%EL(NEL)%NOP(I) ); 
  X(I)  = mGr%nd(Id)%x /1000.0;         ! [m] 
  Y(I)  = mGr%nd(Id)%y /1000.0;         ! [m] 
  Temp_0(i)= mGr%nd(Id)%t;         
 END DO 
 
 DO P=1, mEL4%N_p 

CALL Jacob_2d(J_, J_inv, P, mEL4%N_p, mEL4%NBN, mEL4%N1,mEL4%N2, X, Y, DetJ); 
  T0p=0;  
  DO i=1, mEL4%NBN  
  Ndx(i)=mEL4%N1(i,p)*J_inv(1,1) + mEL4%N2(i,p)*J_inv(1,2) ! [m-1] 

  Ndy(i)=mEL4%N1(i,p)*J_inv(2,1) + mEL4%N2(i,p)*J_inv(2,2) ! [m-1] 
  Ni = mEL4%Nf(i,p); 
      T0p = T0p + Temp_0(i)*Ni; 
  END DO; 
 
  detJ = abs(detJ)*mEL4%W(p);  ! [m2] 
     DO N=1,mEL4%NBN 
   DO I=1,mEL4%NBN    
     Ni = mEL4%Nf(i,p);  
     Nn = mEL4%Nf(n,p); 
 
    Hin = mK*(Ndx(n)*Ndx(i) + Ndy(n)*Ndy(i))*DetJ;  ! Formuła (4.14) 
    Cin = mC*mR*Nn*Ni*detJ;   ! Formuła(5.3) 

est(n,i) = est(n,i) + Hin + Cin/mdTime;          ! Formuła (5.8) 
   r(n)  = r(n) + (Cin/mdTime)*T0p; 
      END DO  
!   Q = 0.8*Tp*Hp 
!   r(n)=r(n) + Nn*Q*DetJ; 
   END DO  
 END DO  
end subroutine PRE_heat_mat; 

Rys. 9.10. Program do obliczenia elementów matrycy sztywności bieŜącego elementu NEL.  

9.3.10. PROCEDURA Z WARUNKAMI BRZEGOWYMI  

Program uwzględniający warunki brzegowe w matryce sztywności elementu pokazano 
na rys. 9.11. W programie dla bieŜącego elementu odbywa się pętla po powierzchniach 
kontaktu tego elementu z otoczeniem (ich liczba jest równa mGr%EL(NEL)%nPov). Dalej 
odbywa się całkowanie po kaŜdej powierzchni za pomocą wybranego w programie IniEL4 

schematu całkowania. W tekście programu podane są odwołania na formuły, wykorzysty-
wane podczas obliczeń dodatków do elementów matrycy i wektora obciąŜeń.  

 
! **************************************************** 
! WARUNKI BRZEGOWE 
! Milenin Andre, AGH, 2008, milenin@agh.edu.pl 
! **************************************************** 
subroutine PRE_heat_pov_mat( NEL ); 
use GlobData; 
implicit none; 
integer*4  :: NEL;  
INTEGER*4  :: I,j,N,P,Id,iPov; 
REAL*8   :: DetJ,Ni,Nn,Pn; 
REAL*8, DIMENSION(4) :: X,Y;   
 DO I=1,4 
  Id = ABS(mGr%EL(NEL)%NOP(I)); 
  X(I)  = mGr%nd(Id)%x /1000.0;         ! [m] 
  Y(I)  = mGr%nd(Id)%y /1000.0;         ! [m] 
 END DO 

 DO iPov=1, mGr%EL(NEL)%nPov 
 id = mGr%EL(NEL)%aPov(iPov); ! Lokalny numer powierzchni elementu 
  SELECT CASE (id) 
  case(1)    
  DetJ=sqrt( (X(4)-X(1))**2 + (Y(4)-Y(1))**2 ); 
  case(2)    
  DetJ=sqrt( (X(1)-X(2))**2 + (Y(1)-Y(2))**2 ); 
  case(3)    
  DetJ=sqrt( (X(2)-X(3))**2 + (Y(2)-Y(3))**2 ); 
  case(4)    
  DetJ=sqrt( (X(3)-X(4))**2 + (Y(3)-Y(4))**2 ); 
  END SELECT  
 
  DO P=1, 2 
   DO n=1, 4 
    ! Pętla po grupom równani (wierszom matrycy) 
    DO i=1, 4  
     ! Pętla po kolumnach matrycy 
    Ni = mEL4%Sf(id)%Nf(i,p);  
    Nn = mEL4%Sf(id)%Nf(n,p);  
   est(n,i) = est(n,i) + mAlfa*Nn*Ni*DetJ;  ! Formuła (4.14) 
    END DO  
   Pn = mAlfa*mT_otoczenia*Nn*DetJ;  ! Formuła (4.15)  
    R(n) = R(n) + Pn; 
   END DO  
  END DO ! P 
 END DO; ! iPov 
end subroutine PRE_heat_pov_mat;  

Rys. 9.11. Program uwzględniający warunki brzegowe w matryce sztywności elementu.  

9.4. PRZYKŁAD OBLICZEŃ 

Wyniki obliczeń dla danych, zamieszczonych na rys. 9.1 są podane na rys. 9.12 dla 
centralnego punktu przekroju (krzywa 1) i na powierzchni (krzywa 2).   


